

\*\*\*\*\*  
UTHSC

PROGRAM NAME :independent variables(patients characteristics  
information-table 1 in proposal)

PROJECT NAME :Finding Equitable and Effective MTM Eligibility  
Criteria

DESCRIPTION :Retrieve information of race, ethnicity, DOB, sex,  
county level information, CCI, and risk adjustment score.

SOFTWARE VERSION :SAS Windows 9.3

RELATED PROGRAMS :N/A

\*\*\*\*\*  
REQUIREMENTS:

\*\*\*\*\*  
Ver# Author & Program History Description Peer reviewer

\*\*\*\*\*  
001 Yanru Qiao production version of the program  
002 Dr. Junling Wang Peer reviewer (Reviewed at September  
27th, 2016)

\*\*\*\*\*;

```
* Output SAS Library *;  
libname ccw 'F:\jwang26\CCW'; * <-- Place Output Library Path Here *;  
libname ccw2 'F:\jwang26\CCW2';  
/*ahrf*/  
proc import datafile="F:\jwang26\Part D data\ahrf for projects using.csv"  
/*AHRF was obtained from: https://data.hrsa.gov/topics/health-workforce/ahrf  
The federal database AHRF provides information on a patient's residence at  
county-level due to unavailability of finer granularity */  
out=ahrf  
dbms=csv  
replace;  
getnames=no;  
guessingrows=32767;  
run;  
  
proc sort data=ahrf;  
by var1;  
run;  
  
data ahrf;  
set ahrf;
```

```

fipscounty=var1; /*ahrf and crosswalk have fipscounty, crosswalk and mbsf have
ssacounty*/
run;

data ccw2.mbsf1;
set ccw2.mbsfsummary2013;
age=BENE_AGE_AT_END_REF_YR; /*referencce year:01/01/2013*/
gender=BENE_SEX_IDENT_CD; /*0 unknown 1 male*/
race=BENE_RACE_CD; /*0 UNKNOWN, 1 WHITE 2 BLACK 3 OTHER 4 ASIAN 5 HISPANIC 6
NORTH AMERICAN NATIVE */
race2=RTI_RACE_CD;
ssacounty=cats(of STATE_CODE BENE_COUNTY_CD);
/*ssacounty=statecode+countycode which was used in crosswalkfile*/
run;

proc sort data=ccw2.mbsf1(keep=BENE_ID age gender race race2 ssacounty);
by ssacounty;
run;

data ahrf2;
merge ccw2.Cbsatocountycrosswalk_fy13(keep=ssacounty fipscounty) ahrf; /*ahrf
and crosswalk have fipscounty to link with Medicare data*/
by fipscounty;
run;

proc sort data=ahrf2;
by ssacounty;
run;

data ccw2.mbsf2;
merge ccw2.mbsf1 ahrf2;
by ssacounty;
run;

proc sort data=ccw2.mbsf2;
by BENE_ID ;
run;

/*independent variables from the AHRF*/
data CCW2.mbsf2b;
set ccw2.mbsf2;
if bene_id ne '';
nonwhite=round((100-VAR6)*0.01, .01);
marriedfamily=round(VAR25/VAR26, .01);
captitaincome=VAR7*1;
POVERTY=round(VAR8*0.01, .01);
education1=round(VAR13*0.01, .01);
EDUCATION2=round(VAR14*0.01, .01);
EDUCATION3=round(VAR15*0.01, .01);
MEDICAID=round(VAR10/VAR9, .01);
UNEMPLOYMENT=round(VAR12*0.01, .01);
NOINSURANCE=round(VAR16*0.01, .01);
if VAR27="1" then msa=1; else msa=0;
if VAR5="Northeast" then region=0;
if VAR5="Midwest" then region=1;
if VAR5="South" then region=2;
if VAR5="West" then region=3;

```

```

if VAR23="1" or var23="2" then hpsa=1; else hpsa=0;
drop var1-var27;
run;

```

```

/*CCI: Identify the presence of each condition in each one of all claims
datasets
Each claim contains the ICD9 codes for the conditions that were diagnosed at
the time. We use those codes to
determine if the patient has each one of 17 conditions included in CCI.
Providers usually report all conditions
identified or affecting the patient at the time of treatment, and not only
the condition that was the primary reason for
the medical encounter. This leads to more accurate and complete description
of a patient's health status. The code
below shows how we identify the conditions in the Inpatient file. The same
logic is applied to all 7 Medicare claims database.*/
data inpatientbase2013;
set ccw2.inpatientbase2013;
array CONDITIONS (17) ANYMI CHF VASCUL1 CVD PULMON1 DEMENTIA PARALYS DIABET1
DIABET3 RENAL1 LIVER1 LIVER2
ANYULCER RHEUM AIDS MALIGNANCY METASTATIC;
do i= 1 to 17; CONDITIONS(i)=0; end;
array dxcode ADMTG_DGNS_CD PRNCPAL_DGNS_CD ICD_DGNS_CD1-ICD_DGNS_CD25;
do over dxcode;
dx_3=substr(dxcode,1,3);
dx_4=substr(dxcode,1,4);
/***** MYOCARDIAL INFARCTION WEIGHT=1 *****/
if dx_3='410' | dx_3='412' then ANYMI=1;
/***** CHF ***** WEIGHT=1 *****/
if dx_3='428' then CHF=1;
/***** PERIPHERAL VASCULAR DISEASE ***** WEIGHT=1***/
if dx_3='441'|dx_4 in('4439','7854','V434','v434') then VASCUL1=1;
/***** CEREBROVASCULAR DISEASE ***** WEIGHT=1 *****/
if '430'<=dx_3<='437'|dxcode='438 ' then CVD=1;
/***** COPD ***** WEIGHT=1 *****/
if '490'<=dx_3<='496'|'500'<=dx_3<='505'|dx_4='5064' then PULMON1=1;
/***** DEMENTIA ***** WEIGHT=1 *****/
if dx_3='290' then DEMENTIA=1;
/***** PARALYSIS ***** WEIGHT=1 *****/
if dx_3='342'|dx_4='3441' then PARALYS=1;
/***** DIABETES ***** WEIGHT=1 *****/
if dxcode='250 '|dx_4='2507'|'2500'<=dx_4<='2503' then DIABET1=1;
/***** DIABETES WITH SEQUELAE ***** WEIGHT=2 *****/
if ('2504'<=dx_4<='2506') then DIABET3=1;
/***** CHRONIC RENAL FAILURE ***** WEIGHT=2 *****/
if dx_3 in('582','585','586','588')| ('5830'<=dx_4<='5837') then RENAL1=1;
/***** VARIOUS CIRRHODITES ***** WEIGHT=1 *****/
if dx_4 in('5712','5714','5715','5716') then LIVER1=1;
/***** MODERATE-SEVERE LIVER DISEASE *** WEIGHT=3***/
if ('5722'<=dx_4<='5728')|('4560'<=dx_4<='4561')|dxcode in('4562
','45620','45621') then LIVER2=1;
/***** ULCERS ***** WEIGHT=1 *****/

```

```

if
('5310'<=dx_4<='5317') | ('5320'<=dx_4<='5327') | ('5330'<=dx_4<='5337') | ('5340'<
=dx_4<='5347') |
dx_4 in('531 ', '5319', '532 ', '5329', '533 ', '5339', '534 ', '5349') then
ANYULCER=1;
/***** RHEUM *****/
if dxcode in('71481', '725 ', '7100 ', '7101 ', '7104 ') | ('7140'<=dx_4<='7142')
then RHEUM=1;
/***** AIDS *****/
if '042'<=dx_3<='044' then AIDS=1;
/***** ANY MALIGNANCY ***/
if ('140'<=dxcode<='1729') | ('174'<=dxcode<='1958') | ('200'<=dxcode<='2089')
then MALIGNANCY=1;
/***** METASTATIC SOLID TUMOR *****/
if '196'<=dxcode<='1991' then METASTATIC=1;
end;
keep bene_id ANYMI CHF VASCUL1 CVD PULMON1 DEMENTIA PARALYS DIABET1 DIABET3
RENAL1 LIVER1 LIVER2
ANYULCER RHEUM AIDS MALIGNANCY METASTATIC;
run;

/*****THE SAME CODES FOR ALL SEVEN CLAIMS DATASETS*****/

data ccw2.cci;
set inpatientbase2013 outpatientbase2013 carrierbase2013 DMEbase2013
hhabase2013 hospicebase2013 snfbase2013;
run;

proc sort data=ccw2.cci nodupkey; by bene_id ANYMI CHF VASCUL1 CVD PULMON1
DEMENTIA PARALYS DIABET1 DIABET3 RENAL1 LIVER1 LIVER2 ANYULCER RHEUM AIDS
MALIGNANCY METASTATIC; run;

/*****count chronic
diseases*****/
data test;
set ccw2.cci(obs=100);
run;

We need to calculate the number of chronic diseases since that is a MTM
eligibility criterion*/
proc means data=ccw2.cci noprint;
class bene_id;
var ANYMI CHF VASCUL1 CVD PULMON1 DEMENTIA PARALYS DIABET1 DIABET3 RENAL1
LIVER1 LIVER2 ANYULCER RHEUM AIDS MALIGNANCY METASTATIC;
output out=ccw2.cci1 max=; /*for each patient. I only need 1 score for each
disease*/
run;
/*proc sort data=ccw2.cci out=test nodupkey ; by bene_id;
run;*/

data ccw2.cci2; set ccw2.cci1;
CCI=ANYMI+CHF+VASCUL1+CVD+PULMON1+DEMENTIA+PARALYS+DIABET1+DIABET3*2+RENAL1*2
+LIVER1+LIVER2*3+ANYULCER+RHEUM+AIDS*6+MALIGNANCY*2+METASTATIC*6;
keep bene_id cci;
if bene_id='' then delete;
run;

```

```

proc sort data=ccw2.cci2 out=cci2;by bene_id;run;
data ccw2.cci3;
merge ccw2.mbsf2b(keep=bene_id) cci2;
by bene_id;
if cci=. then cci=0;
run;

/*****Preparation for risk adjustment score calculation*****/
Centers for Medicare and Medicaid Services (CMS), Department of Health and
Human Services. Risk Adjustment (online). Available at:
https://www.cms.gov/Medicare/Health-Plans/MedicareAdvtgSpecRateStats/Risk-Adjustors.html. Accessed July 12, 2018.*/

/**person--medicaid and new medicaid*****/
/*data ccw2.mbsfd2012;
set ccw.mbsf_d_cmpnts;
DUAL_ELGBL_MOS_NUM2012=DUAL_ELGBL_MOS_NUM;
run;
*/
proc sort data=ccw2.mbsfd2012(keep=bene_ID DUAL_ELGBL_MOS_NUM2012)
out=mbsfd2012;
by bene_id;
run;
proc sort data=ccw2.mbsfd2013(keep=bene_ID DUAL_ELGBL_MOS_NUM) out=mbsfd2013;
by bene_id;
run;
data ccw2.medicaid;
merge mbsfd2012 mbsfd2013;
by bene_id;
run;

/*MCAID-numeric, =1 if number of State Part B BUYIN (MEDICAID)Months of base
year >0,=0 otherwise

EMCAID -numeric, =1 if a new enrollee and number of State Part B BUYIN
(MEDICAID) months of payment year >0; =0 otherwise*/

DATA ccw2.medicaid;
set ccw2.medicaid;
DUAL_ELGBL_MOS_NUM2013=DUAL_ELGBL_MOS_NUM*1;
DUAL_ELGBL_MOS_NUM2012a=DUAL_ELGBL_MOS_NUM2012*1;
if DUAL_ELGBL_MOS_NUM2013>0 then mcaid=1; else mcaid=0;
if DUAL_ELGBL_MOS_NUM2012a=0 and DUAL_ELGBL_MOS_NUM2013>0 then nemcaid=1;
else nemcaid=0;
run;

/*****First file for risk adjustment score: person--PERSON file
• HICNO (or other person identification variable. It must be set in the
macro variable IDVAR) -character or numeric type and unique
• SEX one character, 1=male; 2=female
• DOB- SAS date format, date of birth
OREC
-one character, original reason for entitlement with the following values:

```

```
0 - OLD AGE (OASI)
    1 - DISABILITY (DIB)
    2 - ESRD
    3 - BOTH DIB AND ESRD*/
```

```
data ccw2.person;
set ccw2.mbsfsummary2013;
hicno=bene_id;
sex=BENE_SEX_IDENT_CD*1;
dob=BENE_BIRTH_DT;
orec=BENE_ENTLMT_RSN_ORIG*1;
keep bene_id hicno sex dob orec;
format dob DATE9.;
run;
proc sort data=ccw2.person out=ccw2.person; by bene_id; run;
```

```
data ccw2.person2;
merge ccw2.medicaid(keep=bene_id mcaid nemcaid) ccw2.person;
by bene_id;
run;
```

```
data _null_;
    set ccw2.person2;
    FILE "F:\jwang26\ccw2\person.txt";
    put hicno sex dob mcaid nemcaid orec;
run;
```

```
/******Second File for risk adjustment Score: chronic conditions**
DIAG file--a diagnosis file with at least one record per person-specific
unique diagnosis.
• HICNO (or other person identification variable that must be the same as
in PERSON file)
- person identifier of character or numeric type and unique to an individual

• DIAG

-ICD-9-CM diagnosis code, 5 character field, no periods, left justified. The
user may include all diagnoses or limit the codes to
those used by the model. Codes should be to the greatest level of available
specificity.
Diagnoses should be included only from providers and physician specialties as
provided in prior notices.
*/
/*All of the diagnosis information from 7 claims datasets*/
```

```
data ccw2.d29;
set ccw2.d28;
if diag='' then delete;
run;
```

```
data _null_;
    set ccw2.d29; /*comprehensive diagnosis information*/
    FILE "F:\jwang26\ccw2\diag.txt";
    put hicno diag;
run;
```